## Statistical Programming with SAS/IML ${ }^{\circ}$ Software

## $\Leftrightarrow$ Rick Wicklin

percentage of births by day of the week 0 declare Š́aitėrPlo plot;
blot = Scatter Plot. Wreate(dobj, "Date", "Percentage
olot.SetAvisLabel (YA XIS, A XISLIABEL_ VARIAABEL)
colors = = BLIEE \& / \& Y AN / $/$ YELLOW // RED;
run ColorCodeObsB ${ }^{*}$ Growpor(dobi, "DOW, colo olot.SetMarkerSize ( 7 );

## §sas



## Contents

I Programming in the SAS/IML Language ..... 1
Chapter 1. An Introduction to SAS/IML Software ..... 3
Chapter 2. Getting Started with the SAS/IML Matrix Programming Language ..... 17
Chapter 3. Programming Techniques for Data Analysis ..... 55
Chapter 4. Calling SAS Procedures ..... 89
II Programming in SAS/IML Studio ..... 107
Chapter 5. IMLPlus: Programming in SAS/IML Studio ..... 109
Chapter 6. Understanding IMLPlus Classes ..... 129
Chapter 7. Creating Statistical Graphs ..... 143
Chapter 8. Managing Data in IMLPlus ..... 173
Chapter 9. Drawing on Graphs ..... 187
Chapter 10. Marker Shapes, Colors, and Other Attributes of Data ..... 225
III Applications ..... 249
Chapter 11. Calling Functions in the R Language ..... 251
Chapter 12. Regression Diagnostics ..... 279
Chapter 13. Sampling and Simulation ..... 311
Chapter 14. Bootstrap Methods ..... 349
Chapter 15. Timing Computations and the Performance of Algorithms ..... 371
Chapter 16. Interactive Techniques ..... 385
IV Appendixes ..... 413
Appendix A. Description of Data Sets ..... 415
Appendix B. SAS/IML Operators, Functions, and Statements ..... 419
Appendix C. IMLPlus Classes, Methods, and Statements ..... 427
Appendix D. Modules for Compatability with SAS/IML 9.22 ..... 435
Appendix E. ODS Statements ..... 437
Index ..... 441

From Statistical Programming with SAS/IML ${ }^{\circledR}$ Software. Full book available for purchase here.

## Chapter 2

## Getting Started with the SAS/IML Matrix Programming Language

## Contents

2.1 Overview of the SAS/IML Language ..... 18
2.2 Creating Matrices ..... 18
2.2.1 Printing a Matrix ..... 19
2.2.2 The Dimensions of a Matrix ..... 20
2.2.3 The Type of a Matrix ..... 21
2.2.4 The Length of a Character Matrix ..... 22
2.3 Using Functions to Create Matrices ..... 24
2.3.1 Constant Matrices ..... 24
2.3.2 Vectors of Sequential Values ..... 25
2.3.3 Pseudorandom Matrices ..... 27
2.4 Transposing a Matrix ..... 28
2.5 Changing the Shape of Matrices ..... 29
2.6 Extracting Data from Matrices ..... 30
2.6.1 Extracting Rows and Columns ..... 31
2.6.2 Matrix Diagonals ..... 33
2.6.3 Printing a Submatrix or Expression ..... 35
2.7 Comparision Operators ..... 36
2.8 Control Statements ..... 38
2.8.1 The IF-THEN/ELSE Statement ..... 38
2.8.2 The Iterative DO Statement ..... 39
2.9 Concatenation Operators ..... 41
2.10 Logical Operators ..... 43
2.11 Operations on Sets ..... 46
2.12 Matrix Operators ..... 47
2.12.1 Elementwise Operators ..... 47
2.12.2 Matrix Computations ..... 49
2.13 Managing the SAS/IML Workspace ..... 51

### 2.1 Overview of the SAS/IML Language

SAS/IML is a programming language for high-level, matrix-vector computations. Matrices are rectangular arrays that usually contain numbers. A matrix that contains character data is often explicitly called a character matrix. In statistical programming, matrices often hold data for analysis. Each row of the matrix is an observation; each column of the matrix is a variable.

If your data are in a matrix, you can carry out many statistical operations by using matrix operations. The SAS/IML language has functions and matrix operations that enable you to manipulate matrices as a unit, regardless of the number of rows or columns in the matrix. For an example, see the section "Case Study: Standardizing the Columns of a Matrix" on page 83.

Operations on numerical matrices are also used to describe a wide variety of statistical techniques, including ordinary least squares (OLS) regression and principal component analysis.

This chapter is an introduction to the SAS/IML syntax. It includes basic information about how to define matrices, compare quantities, and call functions and subroutines. It includes a description of basic programming statements such as IF-THEN/ELSE and the iterative DO statement.

### 2.2 Creating Matrices

A matrix is an $n \times p$ array of numbers or character strings. The integers $n$ and $p$ are the dimensions of the matrix. The row dimension is $n$; the column dimension is $p$. A vector is a special case of a matrix. An $n \times 1$ matrix is called a column vector, whereas a $1 \times n$ matrix is called a row vector. A $1 \times 1$ matrix is called a scalar. In general, this book refers to any SAS/IML variable as a matrix, regardless of its dimensions.

In a SAS/IML program, all variables are matrices, so you do not need to specify the type of a variable. Furthermore, matrices are dynamically reassigned as needed, so you do not need to specify the size or the type (numeric or character) of a matrix. For example, the following statements are all valid:

```
/* create matrices of various types and sizes */
x = 1; /* scalar */
x = {1 2 3}; /* reassign to row vector */
y = {1 2 3, 4 5 6}; /* 2 x 3 numeric matrix */
y = {"male" "female"}; /* reassign to 1 x 2 character matrix */
```

See the section "Running a PROC IML Program" on page 7 for instructions on how to run SAS/IML programs. The first statement creates $\mathbf{x}$ as a numerical scalar matrix. The second statement redefines $\mathbf{x}$ as a numerical row vector; spaces separate entries in different columns. The third statement defines $y$ as a $2 \times 3$ numerical matrix; commas indicate a new row. The last statement redefines $y$ as a $1 \times 2$ character matrix.

Programming Tip: When defining a matrix, use a comma to indicate a new row.

### 2.2.1 Printing a Matrix

You can use the PRINT statement to display the value of one or more matrices. The following statement displays the values of the matrices defined in the previous section:

```
print x, y;
```

Figure 2.1 Numeric and Character Matrices


Notice that a comma in the PRINT statement indicates that the second matrix should be displayed on a new row. If you omit the comma, the matrices are displayed side by side.

The PRINT statement has four useful options that affect the way a matrix is displayed:

## COLNAME=matrix

specifies a character matrix to be used for column headings.

## FORMAT=format

specifies a valid SAS or user-defined format to use when printing matrix values.

## LABEL=label

specifies a label for the matrix. If this option is not specified, the name of the matrix is used as a label.

## ROWNAME=matrix

specifies a character matrix to be used for row headings.
These options are specified by enclosing them in square brackets after the name of the matrix that you want to display, as shown in the following example:

```
/* print marital status of 24 people */
ageGroup = {"<= 45", " > 45"}; /* headings for rows */
status = {"Single" "Married" "Divorced"}; /* headings for columns */
counts = 5 5 0, /* data to print */
    2 9 3 };
print counts[colname=status
    rowname=ageGroup
    label="Marital Status by Age Group"];
```

```
pct = counts / 24; /* compute proportions */
print pct[format=PERCENT7.1];
/* print as percentages */
```

Figure 2.2 Matrices Displayed with PRINT Options


You can also use these options by specifying their first letters: $\mathbf{C =}, \mathbf{F}=\mathbf{L}=$, and $\mathbf{R}=$.

### 2.2.2 The Dimensions of a Matrix

You can determine the dimensions of a matrix by using the NROW and NCOL functions, as shown in the following statements:

```
/* dimensions of a matrix */
n_x = nrow (x);
p_x = ncol(x);
print n_x p_x;
```

Figure 2.3 Dimensions of a Matrix

| $n \_x$ | $p_{-} x$ |
| :---: | :---: |
| 1 | 3 |
| $n_{-} u$ | $p_{-} u$ |
| 0 | 0 |

A matrix that contains no elements is called an empty matrix. There are several reasons why a matrix can be empty:

- It has not been defined.
- Its memory was freed by using the FREE statement.
- It is the result of a query that returned the empty set (such as the intersection of disjoint sets).

The output from the following statements (see Figure 2.4) shows that each dimension of an empty matrix is zero.

```
/* dimensions of an empty matrix */
n_u = nrow (empty_matrix);
p_u = ncol(empty_matrix);
print n_u p_u;
```

Figure 2.4 Dimensions of an Empty Matrix

| n_u | p_u |
| ---: | ---: |
| 0 | 0 |

### 2.2.3 The Type of a Matrix

A matrix is either numeric or character or undefined; you cannot create a matrix that contains both numbers and character strings. You can determine whether a matrix is numeric or character by using the TYPE function, as shown in the following statements:

```
/* determine the type of a matrix */
x = {1 2 3};
y = {"male" "female"};
type_x = type(x);
type_y = type(y);
print type_x type_y;
```

Figure 2.5 Types of Matrices

| type_x type_y |  |
| :--- | :--- |
| N | C |

If a matrix is numeric, the TYPE function returns the character ' $N$ '. If a matrix is character, the TYPE function returns the character ' C ' . If a matrix is empty, then the TYPE function returns ' U ' (for "undefined") as shown in the following statements:

```
/* handle an empty or undefined matrix */
type_u = type(undefined_matrix);
if type_u = 'U' then
    msg = "The matrix is not defined.";
else
    msg = "The matrix is defined.";
print msg;
```

Figure 2.6 Result of Handling an Undefined Matrix

| msg |
| :---: | :---: |
| The matrix is not defined. |

Programming Tip: A matrix is either numeric or character or undefined; you cannot create a matrix that contains both numbers and character strings.

### 2.2.4 The Length of a Character Matrix

SAS/IML character matrices share certain similarities with character variables in the DATA step. In the DATA step, the length of a character variable is determined when the variable is initialized: either explicitly by using the LENGTH statement or implicitly by the length of the first value for the variable. Similarly, every element in a SAS/IML character matrix has the same number of characters: the length of the longest element. This length is determined when the matrix is initialized. Strings shorter than the longest element are padded with blanks on the right.

For example, the following statements define a character matrix with length 4 :

```
c = {"Low" "Med" "High"}; /* maximum length is 4 characters */
```

The matrix $c$ is initialized to have length 4, the length of its longest character string. Shorter strings such as "Low" are padded on the right so that the first element of c is stored as Low $\square$ where $\square$ indicates a blank character.

Programming Tip: The length of a character matrix is determined by the length of its longest element at the time it is created.

You can find the length of a character matrix by using the NLENG function, which returns a single number. You can find the number of characters for each element of a character matrix by using the LENGTH function, which returns a number for each element. These functions are shown in the following statements:

```
/* find the length of a character matrix and of each element */
nlen = nleng(c); /* length of matrix */
len = length(c); /* number of characters in each element */
print nlen len;
```

Figure 2.7 Lengths of Elements in a Character Matrix

| nlen | len |  |  |
| ---: | ---: | ---: | ---: | ---: |
| 4 | 3 | 3 | 4 |

Notice that the LENGTH function returns a numerical matrix that is the same dimension as its input argument. For example, the $i j$ th element of len is the number of characters in the $i j$ th element of c. (Strictly speaking, both functions return numbers that represent bytes. Because each English character is one byte long, the numbers also give the number of characters for matrices that contain English strings.)

When you set the value of an element of an existing matrix, the value is truncated if it is too long, as shown in the following statements:

```
/* assign a long string to a matrix with a shorter length */
c[2] = "Medium"; /* value is truncated! */
print c;
```

Figure 2.8 A Truncated Character String


The output from these statements is shown in Figure 2.8. The matrix c was initialized to have length 4 , so when you assign a longer string to an element, only the first four characters fit into the matrix element.

You cannot dynamically change the length of a character matrix, but you can copy its contents to a vector that has a longer (or shorter) character length. The following statements use the PUTC function in Base SAS software to copy a vector of values:

```
/* copy character strings to a matrix with a longer length */
c = {"Low" "Med" "High"}; /* maximum length is 4 characters */
d = putc(c, "$6."); /* copy into vector with length 6 */
d[2] = "Medium"; /* value fits into d without truncation */
nlen = nleng(d);
print nlen, d;
```

Figure 2.9 Result of Changing the Length of a Character Vector


In the previous statements, the PUTC function applies the $\$ 6$. format to every element of $c$. The PUTC function returns a matrix with the same dimensions as c. This matrix is stored into d.

Notice that c contains a vector of character strings, but that the PUTC function acted on each element. This is generally true: you can pass a matrix of values to Base SAS functions and expect them to act on each element.

Programming Tip: You can call functions in Base SAS software from SAS/IML programs. In most cases, these functions act on each element of a matrix.

Strings that are smaller than the maximum length of a character matrix are padded with blanks on
the right. You can see this in Figure 2.9 by noticing that there is more space between the words "Low" and "Medium" than between the words "Medium" and "High." The value stored in the first element of d is "Low $\square \square \square$ " where $\square$ indicates a blank character.

The padding of character strings with blanks on the right can cause a problem when you use the CONCAT function or the string concatentation operator $(+)$ to concatenate strings. The solution to this problem is to use the TRIM function in Base SAS software to remove trailing blanks, as shown in the following statements:

```
/* use the '+' operator to concatenate strings */
msg1 = "I like the " + d[1] + " value."; /* blanks! */
msg2 = "I like the " + trim(d[1]) + " value."; /* no blanks */
print msg1, msg2;
```

Figure 2.10 Result of Concatenating Strings and Removing Trailing Blanks

| msg1 |
| :---: |
| I like the Low value. |
| msg2 |
| I like the Low value. |

There are times when a character string also has leading blanks. You can use the STRIP function in Base SAS software to remove both leading and trailing blanks. In most situations that involve string concatenation, you will want to remove leading and trailing blanks.

Programming Tip: When concatenating character strings, use the STRIP function to remove leading and trailing blanks from elements of a character matrix.

### 2.3 Using Functions to Create Matrices

The SAS/IML matrices in the previous section were created by typing in the elements. More typically, matrices are obtained by using SAS/IML functions to generate data or by reading data from a SAS data set. This section describes creating matrices by using SAS/IML functions; creating matrices by reading a SAS data set is covered in Chapter 3, "Programming Techniques for Data Analysis."

### 2.3.1 Constant Matrices

The simplest matrix is a constant matrix. In SAS/IML, the $\mathbf{J}$ function creates a constant matrix. The syntax is J (nrow, ncol, value), although you can omit the third argument, which defaults to the
value 1 . For example, the following statements create several constant matrices:

```
/* create constant matrices */
c = j(10, 1, 3.14); /* 10 x 1 column vector */
r = j( 1, 5); /* 1 x 5 row vector of 1's */
m = j(10, 5, 0); /* 10 x 5 matrix of zeros */
miss = j(3, 2, .); /* 3 x 2 matrix of missing values */
```

The first statement creates a column vector with ten rows; each element has the value 3.14. The second statement creates a row vector with five columns; each element is a 1 because the value argument is omitted. The third statement creates a $10 \times 5$ matrix of zeros. The last statement creates a $3 \times 2$ matrix in which every element is a missing value.

Programming Tip: Do not confuse an empty matrix with a matrix that contains missing values or with a zero matrix. An empty matrix has no rows and no columns. A matrix that contains missing values has at least one row and column, as does a matrix that contains zeros.

You can also use the REPEAT function to create a constant matrix or, more generally, a matrix with a repeating pattern of values. The REPEAT function creates a new matrix by repeating a given matrix a specified number of times in each dimension, as shown in the following example:

```
/* create a matrix by repeating values from another matrix */
g = repeat ({0 1}, 3, 2); /* repeat the pattern 3 times down */
print g; /* and 2 times across */
```

Figure 2.11 A Repeated Pattern of Values

|  | $g$ |  |  |  |
| :--- | :--- | :--- | :--- | :--- |
|  | 0 | 1 | 0 | 1 |
| 0 | 1 | 0 | 1 |  |

The REPEAT function takes three arguments: a matrix of values, the number of times that matrix should be repeated in the vertical dimension, and the number of times that matrix should be repeated in the horizontal dimension. In the example, the vector $\left\{\begin{array}{ll}0 & 1\end{array}\right\}$ is repeated three times down the rows and twice across the columns.

You can also use the $\mathbf{J}$ function and the REPEAT function to create character matrices.

### 2.3.2 Vectors of Sequential Values

The next simplest matrix to construct is a matrix in which entries follow an arithmetic progression. The DO function (not to be confused with the iterative DO statement) creates a vector with elements that follow an arithmetic series. The syntax is DO (start, stop, increment). Similar to the DO function is the "colon operator," which can create an arithmetic series with an increment of 1 or -1 .

## 26

(The colon operator is also known as the index creation operator since it is often used to create an index start:stop.) The following statements demonstrate these functions:

```
/* create a vector of sequential values */
i = 1:5; /* increment of 1 */
j = 5:1; /* increment of -1 */
k = do(1, 10, 2); /* odd numbers 1, 3, ..., 9 */
print i, j, k;
```

Figure 2.12 Vectors with Sequential Values

|  | i |  |  |  |  |
| :---: | :---: | :---: | :---: | :---: | :---: |
| 1 | 2 |  | 3 | 4 | 5 |
|  |  | j |  |  |  |
| 5 | 4 |  | 3 | 2 | 1 |
|  |  | k |  |  |  |
| 1 | 3 |  | 5 | 7 | 9 |

The index creation operator (:) has lower precedence than arithmetic operators, as shown by the following statements and by Figure 2.13:

```
/* the index creation operator has low precedence */
n1 = 1;
n2 = 10;
h = n1+2:n2-3; /* equivalent to 3:7 */
print h;
```

Figure 2.13 Precedence of Index Creation Operator

|  |  | h |  |
| :--- | :--- | :--- | :--- | :--- | :--- | :--- |

Programming Tip: The index creation operator (:) has lower precedence than arithmetic operators. Therefore, $a+b * c: d+e * f$ is equivalent to $(a+b * c):(d+e * f)$.

The DO function requires that its arguments are numerical. But there is one situation in which the index creation operator can be used with character values: the creation of variable names with a common prefix and a numerical suffix. For example, if you have 10 variables and want to name them $x 1, x 2, \ldots, x 10$, the index creation operator can create these variable names, as shown in the following statements:

```
/* create variable names with sequential values */
varNames = "x1":"x10";
print varNames;
```

Figure 2.14 Variable Names with Sequential Values


Programming Tip: Use the index creation operator to create a vector of names with a common prefix and a numerical suffix (for example, "x1": "x10").

### 2.3.3 Pseudorandom Matrices

Probability theory and statistics describe events that contain aspects of randomness. A random number algorithm is an algorithm that generates a sequence of numbers whose statistical properties are such that the sequence is indistinguishable from a truly random sequence. Of course, it is technically impossible to generate a random sequence, since, by definition, a random process is not deterministic. Therefore, some people prefer use the term pseudorandom numbers to describe a sequence of numbers that are generated by a computer and that behave similarly to random variates from a specified distribution.

In SAS software a pseudorandom sequence of numbers is initialized with a seed value that determines the sequence. If you use a different seed value, you get a different sequence of numbers.

There are several SAS/IML functions and subroutines that generate pseudorandom variates. The UNIFORM and NORMAL functions generate random variates from the uniform distribution on $[0,1]$ and from the standard normal distribution, respectively. For example, the following statements generate two variables that are linearly related to each other.

```
/* create pseudorandom vectors */
seed = j(10, 1, 1); /* set seed (=1) and dimensions */
x = uniform(seed); /* 10 x 1 pseudorandom uniform vector */
y = 3*x + 2 + normal(seed); /* linear response plus normal error */
```

The UNIFORM and NORMAL functions are convenient for simple simulations and for quickly generating test data. In the previous statements, the seed value is 1 . The size and shape of the seed matrix determines the shape of the output from the UNIFORM and NORMAL functions. Since seed is a $10 \times 1$ vector, the column vector $\mathbf{x}$ contains 10 pseudorandom numbers in the interval $[0,1]$. Similarly, the NORMAL function returns a normally distributed "error vector," so that the vector $\mathbf{y}$ is a linear function of $\mathbf{x}$ plus an error term.

Programming Tip: Use the UNIFORM and NORMAL functions when you need to quickly generate a small random sample for an example or for testing purposes.

The statistical properties of the pseudorandom numbers generated by the UNIFORM and NORMAL functions are not as good as those generated by the newer Mersenne-Twister random number
generator that is implemented in the RANDGEN subroutine. Consequently, you should use the RANDGEN subroutine when you intend to generate millions of pseudorandom numbers.

The RANDGEN subroutine is used extensively in Chapter 13, "Sampling and Simulation." When you use RANDGEN, you need to allocate a matrix that will hold the random numbers, perhaps by using the J function. For the sake of completeness, the following statements use the RANDGEN subroutine to generate two variables that are linearly related to each other:

```
/* create pseudorandom vectors (better statistical properties) */
call randseed(12345); /* set seed for RANDGEN */
x = j(10, 1); /* allocate 10 x 1 vector */
e = x; /* allocate 10 x 1 vector */
call randgen(x, "Uniform"); /* fill x; values from uniform dist */
call randgen(e, "Normal"); /* fill e; values from normal dist */
y = 3*x + 2 + e; /* linear response plus normal error */
```

Programming Tip: Use the RANDGEN subroutine when the statistical properties of the pseudorandom numbers are important or when you intend to generate millions of variates.

For more information about the numerical properties of SAS routines that generate pseudorandom numbers, see the section "Using Random-Number Functions and CALL Routines" in the SAS Language Reference: Dictionary.

### 2.4 Transposing a Matrix

The vectors created by the DO function and the index creation operators are row vectors. To get a column vector, you can use the T function, which transposes a matrix. The syntax is T (matrix).

The transpose of a row vector is a column vector, and the transpose of a column vector is a row vector. The transpose of a two-dimensional matrix is obtained by flipping the matrix about its main diagonal. Specifically, if $a_{i j}$ is the element in the $i$ th row and $j$ th column of a matrix $A$, then $a_{j i}$ is the element in the $i$ th row and $j$ th column of the transpose of $A$. The following statements demonstrate the transpose function:

```
/* transpose a matrix */
s = {1 2 3, 4 5 6, 7 8 9, 10 11 12}; /* 4 x 3 matrix */
transpose = t(s); /* 3 x 4 matrix */
print transpose;
```

Figure 2.15 A Transposed Matrix

| transpose |  |  |  |  |
| :---: | :---: | :---: | :---: | :---: |
|  | 1 | 4 | 7 |  |
| 2 | 5 | 8 | 10 |  |
| 3 | 6 | 9 | 11 |  |

A mathematical notation for the transpose of a matrix $A$ is $A^{\prime}$. The SAS/IML language also supports this syntax:

```
sPrime = s`; /* alternative notation to transpose a matrix */
```

The transpose operator is described in Section 2.12.

### 2.5 Changing the Shape of Matrices

Sometimes it is convenient to reshape the data in a matrix. Suppose you have a $1 \times 12$ matrix. This same data can fit into many other matrices: for example, a $2 \times 6$ matrix, a $3 \times 4$ matrix, a $4 \times 3$ matrix, and so on. The SHAPE function enables you to specify the number of rows and columns for a new matrix. The values for the new matrix come from an existing matrix, as shown in the following statements:

```
/* reshape a matrix */
```

$\mathrm{x}=1: 12$; /* 1 x 12 matrix */
$\mathrm{s}=$ shape $(\mathrm{x}, 4,3)$; $/ *$ reshape data into $4 \times 3$ matrix */

The matrix $\mathbf{s}$ is identical to the one specified manually in the example in the preceding section.
The data in SAS/IML matrices are stored in row-major order and this ordering of the elements is used in reshaping the data. The SHAPE function does not change the order of the data elements in memory; it merely changes how those data are interpreted as a matrix.

Programming Tip: Data in SAS/IML matrices are stored in row-major order.

The previous example specifies both the number of rows and the number of columns to the SHAPE function. You can also specify only the number of rows or only the number of columns. The dimension that is not specified is determined automatically by dividing the number of elements in the matrix by the number of specified rows or columns. To specify only the number of rows, omit the third argument or use 0 for the number of columns. To specify only the number of columns, specify 0 for the number of rows, as shown in the following statements:

```
/* omit dimension; automatically determined */
s1 = shape (x, 4); /* 4 rows ==> 3 columns */
s2 = shape (x, 0, 3); /* 3 columns ==> 4 rows */
```

You can also reshape data into dimensions that are not congruent to the original data by specifying a value to use when the original data "runs out," as shown in the following statements:

```
/* pad data with a specified value (missing) */
s = shape(1:10, 4, 3, .); /* 4 x 3, pad with missing value */
print s;
```

Figure 2.16 Result of the SHAPE Function

| s |  |  |  |
| :---: | :---: | :---: | :--- |
| 1 |  | 2 | 3 |
| 4 | 5 | 6 |  |
| 7 | 8 | 9 |  |
| 10 | $\cdot$ | $\cdot$ |  |

### 2.6 Extracting Data from Matrices

This section describes various techniques for extracting or modifying the subsets of a matrix. The subsets include, rows, columns, submatrices, and diagonal elements.

You can specify elements of a matrix with one index or with two. For example, $\mathbf{x}[3]$ is the third element of the matrix $\mathbf{x}$, specified in row-major order. Similarly, $\mathbf{y}[1,2]$ specifies the element in the first row and the second column of the matrix $\mathbf{y}$. You can use indices on either side of an assignment statement, as the following statements indicate:

```
/* use indices on either side of assignment statements */
x = {1 2 3, 4 5 6};
y = x[2, 3]; /* value of 2nd row, 3rd column */
x[2, 3] = 7; /* changes the value of x[2,3]; y is unchanged */
print x, y;
```

Figure 2.17 Result of Assignment Statements

| x |  |  |
| :---: | :---: | :---: |
| 1 | 2 | 3 |
| 4 | 5 | 7 |
| y |  |  |
|  | 6 |  |

If m is an $n \times p$ matrix, it is an error to refer to $\mathrm{m}[\mathrm{i}, \mathrm{j}]$ when $i>n$ or $j>p$. It is also an error to use nonpositive indices.

It is an error to refer to a subscript of an undefined matrix. For example, the following statements show that you cannot assign elements of a vector if the vector has not been created:

```
/* ERROR: subscripting a matrix that has not been created */
z[1] = 0; /* z is undefined */
```

The error message for this mistake is displayed in the SAS log, which is shown in Figure 2.18.

Figure 2.18 Error Message for Undefined Matrix

```
ERROR: (execution) Matrix has not been set to a value.
    operation : [ at line 1520 column 2
    operands : z, *LIT1001, *LIT1002
z 0 row (type ?, size 0)
*LIT1001 1 row 1 col (numeric)
    1
*LIT1002 1 row 1 col (numeric)
    0
statement : ASSIGN at line 1520 column 1
```

Each line of the error message gives information about the error:

- The error is that you are trying to access an element of a matrix that has not been assigned a value.
- The error occurred in the left-bracket (subset) operation.
- The operation involved three operands: the matrix $\mathbf{z}$ and two unnamed literals.
- The matrix $\mathbf{z}$ has zero rows and zero columns. It is an undefined matrix.
- The first unnamed literal has the value 1.
- The second has the value 0 .
- The error occurred during the assignment statement.

Programming Tip: It is an error to refer to a subscript of an undefined matrix.

### 2.6.1 Extracting Rows and Columns

You can specify submatrices of a matrix by specifying vectors of indices. For example, the following statements assign the matrix $\mathbf{w}$ to the values in the odd rows and even columns of the matrix z:

```
/* extract submatrix */
z={1 2 3 4, 5 6 7 8, 9 10 11 12};
rows = {1 3};
cols = {2 4};
w = z[rows, cols];
print w;
```

Figure 2.19 A Submatrix

| w |  |
| :---: | :---: |
|  | 4 |
| 2 | 4 |
| 10 | 12 |

You can specify all rows of a matrix by omitting the row index. Columns are handled similarly. The following statements use the previous definitions of rows and columns to show this syntax:

```
/* extract only rows or columns */
oddRows = z[rows, ]; /* specified rows; all columns */
evenCols = z[ , cols]; /* all rows; specified columns */
```

Programming Tip: You can specify all rows of a matrix by omitting the row index. Columns are handled similarly.

When you extract a subset of a matrix by specifying a single set of indices, the resulting matrix is always a column vector. At times, this is a surprising result, as shown in the following statements.

```
/* different ways to specify elements of a matrix */
z = {1 3 5 7 9 11 13}; /* row vector */
w = z[{2 4 6}]; /* column vector with values {3,7,11} */
t = z[ , {2 4 6}]; /* row vector with values {3 7 11} */
```

print $w t$;

Figure 2.20 Results of Two Indexing Schemes

| $w$ | $t$ |  |  |
| :---: | :---: | :---: | :---: |
| 3 | 3 | 7 | 11 |
| 7 |  |  |  |

Note that $\mathbf{w}$ is a column vector even though $\mathbf{z}$ is a row vector! To get a row vector, you must explicitly omit the row index, as for the vector t (or use the SHAPE function).

Programming Tip: If you extract a subset of a matrix by specifying a single set of indices, the resulting vector is a column vector.

You can also extract all rows except those that you enumerate. The SETDIF function (described in Section 2.11) is useful for this. The following statements extract all rows except those contained in the vector v :

```
/* extract all rows except those specified in a vector v */
q = {1 2, . 3, 4 5, 6 7, 8 .};
v={2 5}; /* specify rows to exclude */
idx = setdif(1:nrow(q), v); /* start with 1:5, exclude values in v */
r = q[idx, ]; /* extract submatrix */
print idx, r;
```

Figure 2.21 Rows That Are Not Excluded


In the previous statements, the second and fifth rows of the $q$ matrix contain missing values. Suppose you want to exclude these rows from a computation. You can define $\mathbf{v}$ to be the vector that contains the rows to exclude and use the SETDIF function to remove those rows from the vector of all row numbers in $\mathbf{q}$. The vector $\mathbf{i d x}$ contains the rows that are retained, as shown in Figure 2.21.

### 2.6.2 Matrix Diagonals

An important subset of a matrix is the diagonal of the matrix. For an $n \times p$ matrix $A$, the diagonal is the set of elements $A_{i i}$ for $i=1, \ldots, \min (n, p)$. You can extract the diagonal of a matrix by using the VECDIAG function, as shown in the following example:

```
/* extract matrix diagonal */
m = {1 2 3,
    2 5 6,
    3 6 10};
d = vecdiag(m);
print d;
```

Figure 2.22 The Diagonal of a Matrix


The VECDIAG function returns a vector of diagonal elements from a matrix, whereas the DIAG function returns a diagonal matrix created from a specified vector of values, as shown in the following example:

```
/* create a diagonal matrix from a vector */
vals = {5, 2, -1};
s = diag(vals);
print s;
```

Figure 2.23 A Diagonal Matrix

| $\mathbf{s}$ |  |  |  |
| :--- | :--- | :--- | :--- |
|  | 5 | 0 | 0 |
| 0 | 2 | 0 |  |
| 0 | 0 | -1 |  |

The I function returns a square matrix that contains ones on the diagonal. This is called the identity matrix. You need to specify the dimension of the matrix, as shown in the following example:

```
/* create an identity matrix */
ident = I(3); /* 3 x 3 identity matrix */
print ident;
```

Figure 2.24 An Identity Matrix

| ident |  |  |  |
| :--- | :--- | :--- | :--- |
|  | 1 | 0 | 0 |
| 0 | 1 | 0 |  |
| 0 | 0 | 1 |  |

If you want to extract or modify the diagonal values of a general $n \times p$ matrix, you can directly index the elements of the diagonal by using the following statements:

```
/* index the diagonal elements of a matrix */
n = nrow (m);
p = ncol (m);
diagIdx = do(1, n*p, p+1); /* indices of the diagonal */
print diagIdx;
d2 = m[diagIdx]; /* extract the diagonal */
print d2;
```

Figure 2.25 Indices and Values of a Matrix Diagonal


In the example, the DO function constructs the indices of the diagonal for a general $n \times p$ matrix. These are the correct indices because SAS/IML matrices are stored in row-major order. The indices are stored in the diagIdx vector.

You can use the indices in diagIdx to extract the diagonal, as shown in the previous example, or to assign to the diagonal elements. For example, a common matrix operation is to subtract a constant from the diagonal of a square matrix, as shown in the following statements:

```
/* compute B = (m - lambda*I) for lambda=1 */
/* First approach: use the formula */
B1 = m - I(n); /* I(n) gives n x n identity matrix */
/* Second approach: Do not physically create the identity matrix */
B = m;
B[diagIdx] = m[diagIdx] - 1; /* modify the diagonal */
print m B;
```

Figure 2.26 Matrix with Modified Diagonal

| m |  |  | B |  |  |
| :--- | :--- | ---: | :--- | :--- | :--- |
|  |  | 2 | 3 | 0 | 2 |
| 3 | 5 | 6 | 2 | 4 | 6 |
| 2 | 6 | 10 | 3 | 6 | 9 |
|  |  |  |  |  |  |

The example creates the matrix $m-I$ in two different ways. The first way is to use the I function to explicitly create an identity matrix and then to subtract that matrix from $m$ to form the matrix $B$. This works, but is not as efficient as the alternative approach. The alternative approach initializes the matrix $B$ with the values of $m$ and then subtracts 1 from each diagonal element.

### 2.6.3 Printing a Submatrix or Expression

There are times when it is convenient to print a submatrix of a matrix, or, in general, a temporary result of some matrix expression. You can print submatrices and expressions by enclosing the term that you want to print in parentheses.

For example, suppose that a data matrix contains hundreds of rows, but you want to print only the first few rows. You can index the rows that you want to print, and enclose the expression in parentheses, as shown in the following statements:

```
x = shape(1:1000, 0, 4); /* 4 columns, hundreds of rows */
print (x[1:3,]); /* print first three rows */
```

Figure 2.27 First Three Rows of a Matrix

| 1 | 2 | 3 | 4 |
| ---: | ---: | ---: | ---: |
| 5 | 6 | 7 | 8 |
| 9 | 10 | 11 | 12 |

The output is shown in Figure 2.27. Notice that the output does not contain any matrix names. This is in contrast to printed output in previous examples in which the name of the matrix is printed in the output. The explanation for this is that when SAS/IML encounters an expression enclosed in parentheses, it creates a temporary matrix to hold the result and the PRINT statement does not output a name for temporary matrices. However, as described in section "Printing a Matrix" on page 19, you can use the LABEL= and COLNAME= options in the PRINT statement to make the printed output easier to understand, as shown in the following statements:

```
varNames = 'Col1':'Col4';
print (x[1:3,])[label="My Data" colname=varNames];
```

Figure 2.28 Adding a Label and Column Names to Printed Output

| My Data |  |  |  |
| :---: | :---: | ---: | ---: |
| Col1 | Col2 | Col3 | Col4 |
|  |  |  |  |
| 1 | 2 | 3 | 4 |
| 5 | 6 | 7 | 8 |
| 9 | 10 | 11 | 12 |

In the same way, you can print matrix expressions by enclosing the expression in parentheses. For example, the following statements print a linear transformation of a row vector:

```
x = 1:4;
print (3*x + 1); /* print expression */
```

Figure 2.29 Result of Printing an Expression

| 4 | 7 | 10 | 13 |
| :--- | :--- | :--- | :--- |

Programming Tip: To print a submatrix, you must enclose the submatrix in parentheses:

```
print (x[1:3]);
```

This is a peculiarity of the PRINT statement. You also need to enclose matrix expressions in parentheses:

```
print (3*x + 1);
```


### 2.7 Comparision Operators

In the SAS/IML language, the equal sign (=) plays two roles. The equal sign can be an assignment operator or a comparison operator. The other comparison operators are less than ( $<$ ), less than or equals ( $<=$ ), greater than ( $>$ ), greater than or equals ( $>=$ ), and the not equals operator $(\wedge=$ ). Usually you will compare a matrix with a scalar value or with another matrix of the same dimensions, although it is also possible to compare a matrix with a vector. The result of the comparison is a
matrix of zeros and ones. The result matrix has the value one for locations where the comparison is true and the value zero for locations where the comparison is false. This is shown in the following statements:

```
/* comparison operators */
x = {1 2 3, 2 1 1};
s1 = (x=2);
print s1;
z={1 2 3, 3 2 1};
s2 = (x<z);
print s2;
```

Figure 2.30 Results of Comparison Operators

|  | s1 |  |  |
| :---: | :---: | :---: | :---: |
| 0 | 0 | 1 | 0 |
| 1 | 1 | 0 | 0 |
|  | s2 |  |  |
| 0 | 0 | 0 | 0 |
| 1 | 1 | 1 | 0 |

The matrix $\mathbf{s} 1$ has the same dimensions as the matrix $\mathbf{x}$. The element $\mathbf{s} 1[i, j]$ has the value 1 when $\mathbf{x}[i, j]$ equals 2 . Similarly, the matrix $\mathbf{s 2}$ has the value 1 in locations where $\mathbf{x}[i, j]$ is less than $z[i, j]$, and has the value 0 otherwise.

The parentheses in the previous example are not necessary, but might help to remind you that the first equal sign is an assignment, whereas the second is a comparison operator. The SAS/IML language does not support the "multiple assignment" syntax found in some other languages. The expression $\mathbf{x}=\mathbf{y}=0$ does not assign the value zero to the matrices $\mathbf{x}$ and $\mathbf{y}$; instead, it compares the matrix $\mathbf{y}$ with zero, and assigns the result to $\mathbf{x}$.

Programming Tip: The comparison operators return a matrix of zeros and ones.

Programming Tip: Unlike the DATA step, the SAS/IML language does not support the mnemonic keywords EQ, NE, GT, LT, GE, or LE as a replacement for the symbols $=, \wedge=,>,<,>=$, or $<=$.

The comparison operators treat a missing value as a value that is less than any valid nonmissing value, as shown in the following example:

```
/* missing values compare as less than any nonmissing value */
m = .;
n = 0;
r = (m<n);
print r;
```

Figure 2.31 Result of Comparing a Missing Value
$\square$

### 2.8 Control Statements

This section describes SAS/IML statements that control the flow of a program. This includes the IF-THEN/ELSE statements, the iterative DO statement, the DO/WHILE statement, and the DO/UNTIL statement.

### 2.8.1 The IF-THEN/ELSE Statement

The comparison operators are most often used in the conditional expression of an IF-THEN/ELSE statement, as shown in the following statements:

```
/* an IF-THEN/ELSE statement */
x = {1 2 3, 2 1 1};
z = {1 2 3, 3 2 1};
if }x<= z the
    msg = "all(x <= z)";
else
    msg = "some element of x is greater than the corresponding element of z";
print msg;
```

Figure 2.32 Result of an IF-THEN/ELSE Statement
$\square$

The syntax is identical to that used in the DATA step, except that the conditional expression in the SAS/IML language can be a matrix. Recall that the expression $\mathbf{x}<=\mathbf{z}$ resolves to a matrix of zeros and ones. If every element of $\mathbf{x}<=\mathbf{z}$ is nonzero (that is, the expression is true for all elements), then the statement following the THEN keyword is executed. Otherwise, the statement following the ELSE keyword is executed. The ELSE statement is optional.

Programming Tip: A matrix expression in an IF-THEN statement is true provided that all elements of the matrix expression are nonzero.

You can use the DO and END statements to group multiple statements that should be executed, as shown in the following statements:

```
/* a DO-END block of statements */
if x <= z then do;
    msg = "all(x <= z)";
    /* more statements ... */
end;
else do;
    msg = "some element of x is greater than the corresponding element of z";
    /* more statements ... */
end;
```

You can use the ALL function to emphasize the condition you are testing:

```
/* the ALL statement */
if all(x <= z) then do;
    msg = "all(x <= z)";
    /* more statements ... */
end;
```

The ALL function returns the value 1 if every element of its argument is nonzero; otherwise, it returns the value 0 . If you want to test whether any element in a matrix is nonzero, you can use the ANY function, as shown in the following statements:

```
/* the ANY statement */
if any(x < z) then
    msg = "some element of x is less than the corresponding element of z";
print msg;
```

Figure 2.33 Result of the ANY Function

```
        msg
    some element of }x\mathrm{ is less than the corresponding element of }
```

The ANY function returns 1 if any element of its argument is nonzero; otherwise, it returns 0.

### 2.8.2 The Iterative DO Statement

The iterative DO statement enables you to repeat a group of statements several times. For example, you can loop over elements in a matrix or loop over variables in a data set. The syntax is identical to that used in the DATA step, as shown in the following statements:

```
/* the iterative DO statement */
x = 1:5;
sum = 0;
do i = 1 to ncol(x);
    sum = sum + x[i]; /* inefficient way to sum elements */
end;
print sum;
```

The statements loop over the columns of $\mathbf{x}$ and add up each entry. The partial sum at each step in the iteration is accumulated in the variable sum. When the loop ends, sum contains the sum of the elements in $\mathbf{x}$, as shown in Figure 2.34.

Figure 2.34 Result of the Iterative DO Statement

```
sum
```

In general, you should try to avoid looping over every element in a vector. The SAS/IML language has many functions and statements that enable you to avoid explicit loops. For example, you can rewrite the previous example by using the SUM function to eliminate the loop:

```
x = 1:5;
sum = sum(x); /* efficient; eliminate DO loop */
```

The SUM function and other functions that act on matrices are described in Appendix C. The section "Writing Efficient SAS/IML Programs" on page 79 discusses other ways to avoid loops.

The iterative DO statement supports an optional WHILE or UNTIL clause. You can use the WHILE or UNTIL clauses when you want to exit the DO loop after a certain criterion is satisfied, as shown in the following statements:

```
/* an UNTIL clause */
x = 1:5;
sum = 0;
do i = 1 to ncol(x) until(sum > 8);
    sum = sum + x[i];
end;
print sum;
```

Figure 2.35 Result of the UNTIL Clause

| sum |  |
| :--- | :--- |
|  |  |
|  | 10 |

The WHILE clause is evaluated at the top of the loop, whereas the UNTIL clause is evaluated at the bottom of the loop. Consequently, you can use the WHILE clause when you want to block the execution of the body of the loop. For example, the following statements find the partial sum of a vector until a missing value is encountered:

```
/* a WHILE clause */
x = {1 2 . 4 5};
sum = 0;
do i = 1 to ncol(x) while(x[i]^=.);
    sum = sum + x[i];
end;
print sum;
```

Figure 2.36 Result of the WHILE Clause


The statements work correctly because the WHILE clause forces the loop to exit as soon as $\mathbf{i}$ is incremented to 3 . If you try to use an UNTIL clause (such as until ( $x[i]=$. ), the body of the loop executes and results in the missing value being added to the sum variable.

You can also use the WHILE and UNTIL clauses in conjunction with a noniterative DO statement. In this case, you need to control the iteration yourself, and you need to avoid conditions that might lead to infinite loops. For example, the following statements use a DO-UNTIL statement to approximate the first local maximum of the sine function for $x>0$ :

```
/* a DO-UNTIL statement */
x = 0;
dx = 0.01;
do until(sin(x) > sin(x+dx));
    x = x + dx;
end;
print x;
```

Figure 2.37 Result of the DO-UNTIL Statement
$\square$

The true value of the local maximum is $\pi / 2 \approx 1.57$. The loop adds a small increment to $\mathbf{x}$ at each iteration. The condition in the UNTIL clause tests whether the sine function is increasing at the current value of $\mathbf{x}$. The loop continues until $\mathbf{x}$ attains a value at which the sine function is no longer increasing.

### 2.9 Concatenation Operators

A previous section described ways to extract a subset of a matrix. This section describes how you can concatenate matrices together to form a bigger matrix from smaller ones.

The SAS/IML language provides two concatenation operators. The horizontal concatenation operator (1I) appends new columns to a matrix, or combines two matrices that have the same number of rows. A typical usage is to create a data matrix consisting of columns from several vectors, as shown in the following statements:

```
/* horizontal concatenation */
a = {1,2,3,4,5}; /* 5 x 1 column vectors */
b = {3,5,4,1,3};
c = {0,1,0,0,1};
x = a || b || c; /* 5 x 3 matrix */
print x;
```

Figure 2.38 Result of Horizontal Concatenation

|  | $\mathbf{x}$ |  |
| :--- | :--- | :--- |
|  |  |  |
|  |  |  |
| 1 | 3 | 0 |
| 2 | 5 | 1 |
| 3 | 4 | 0 |
| 4 | 1 | 0 |
| 1 | 3 | 1 |

The vertical concatenation operator (//) appends new rows or combines matrices that have the same number of columns. A typical usage is that you want to combine several subsets of data, where each subset is contained in its own matrix. For example, the following statements combine a subset of the data (say, for males) with another subset (say, for females):

```
/* vertical concatenation */
males = {1 3 0, 2 5 1, 3 4 0}; /* 3 x 3 matrix */
females = {4 1 0, 5 3 1}; /* 2 x 3 matrix */
x = males // females; /* 5 x 3 matrix */
```

Because the concatenation operators allocate a new matrix and copy data into it, you should try to avoid using concatenation operators inside of a loop. For example, suppose you want to construct a vector of even integers $\left\{\begin{array}{llll}2 & 4 & 6 & 10\end{array}\right\}$. The following statements show an inefficient way to construct this vector:

```
/* construct vector of even integers */
free x; /* make sure x is empty */
do i = 1 to 5;
    x = x | 2*i; /* inefficient! 5 allocations */
end;
```

The program first uses the FREE statement to ensure that $\mathbf{x}$ is an empty matrix. Inside the DO loop, the program allocates a new matrix for each iteration of the loop. The first iteration concatenates the empty matrix with the scalar that contains 2 . At the end of the first iteration, the matrix $\mathbf{x}$ is $1 \times 1$. At the end of the second iteration, the matrix $\mathbf{x}$ is $1 \times 2$, and so on.

It is not merely the allocation of memory that is inefficient. The algorithm is also inefficient because of the way it copies elements multiple times. During the five iterations, the value 2 is copied into a new matrix five times. The value 4 is copied four times, and so on. In total, the five values are copied 15 times before the loop ends. If the same algorithm is used to create a vector of length $n$, the values are copied a total of $n(n-1) / 2$ times!

If the purpose of the loop is to compute values for a matrix with a given number of rows and columns, it is usually more efficient to allocate space for the matrix prior to the loop. You can assign the values to the preallocated matrix inside the loop, as shown in the following statements:

```
/* Improved algorithm: allocate the matrix to hold the
    even integers. Assign values into the vector. */
x = j(1, 5, .); /* allocate; fill with missing */
do i = 1 to ncol(x);
    x[i] = 2*i; /* assign value */
end;
```

This algorithm requires a single allocation of a vector, and no values are copied unnecessarily. Of course, for this simple example, you can avoid the DO loop altogether:

```
x = 2*(1:5);
```

Programming Tip: Whenever possible, avoid using concatenation operators to construct a matrix row-by-row (or column-by-column) inside a loop.

You cannot always avoid concatenation within a DO loop because there are situations in which you do not know in advance how many iterations are required. Nevertheless, even in those situations you can often obtain an upper bound on the number of iterations and preallocate the results matrix. An example is a root-finding algorithm that iterates until convergence. You do not know in advance how many iterations are required, but you can decide to limit the algorithm to no more than 50 iterations before stopping the algorithm. In this case, you can preallocate space for up to 50 results.

The concatenation operators enable you to append one matrix to the end of another. If you need to insert rows or columns into the middle of a matrix, you can use the INSERT function. If you need to remove rows or columns from the middle of a matrix, you can use the REMOVE function. For example, the following statements remove the first, third, and sixth elements of a vector, which has the effect of removing all missing values from the vector:

```
y = {., 1, ., 2, 3, .};
v = remove(y, {1 3 6});
```

See the SAS/IML User's Guide for details of the INSERT and REMOVE statements.

### 2.10 Logical Operators

The IF-THEN/ELSE, DO-UNTIL, and DO-WHILE statements all test whether a condition is true. When the condition involves a matrix, the condition is considered "true" provided that it is true for each element of the matrix.

You can also test whether multiple conditions are true and combine two or more conditions into a single logical expression. The three logical operators are the AND operator (\&), the OR operator $(I)$, and the NOT operator ( ${ }^{\wedge}$ ). The operators perform logical operations on the elements of matrices, as shown in the following statements:

```
/* logical operators */
r = {0 0 1 1};
s = {0 1 0 1};
and = (r & s);
or = (r | s);
not = ^r;
print and, or, not;
```

Figure 2.39 Results of Logical Operators

|  | and |  |  |  |
| :---: | :---: | :---: | :---: | :---: |
|  |  |  |  |  |
|  | 0 | 0 | 0 | 1 |
|  |  |  |  |  |
|  | 0 | 1 | 1 | 1 |

As the example indicates, the matrix and contains a nonzero value only where the corresponding elements of $\mathbf{r}$ and $\mathbf{s}$ are nonzero. Similarly, the matrix or contains a nonzero value if either of the corresponding elements of $\mathbf{r}$ and $\boldsymbol{s}$ are both nonzero. Lastly, the matrix not is nonzero only where the matrix $r$ is zero.

Programming Tip: Unlike the DATA step, the SAS/IML language does not support the mnemonic keywords AND, OR, or NOT as a replacement for the symbols $\&$, I, and $\wedge$.

You can use the logical operators to combine multiple criteria into a single criterion. For example, the following statements use the logical AND operator to test whether the values of a vector are all in the domain of a function:

```
/* logical combination of criteria */
x = do(-1, 1, 0.5);
if (x>= -1) & ( }x<=1\mathrm{ ) then
    y = sqrt(1-x##2); /* the ## operator squares each element of x */
else
    y = "Unable to compute the function.";
print y;
```

Figure 2.40 Result of the Logical AND (\&) Operator

## y

$0 \quad 0.8660254$
10.8660254

0

There are two conditions in the IF-THEN statement. The first tests whether all of the elements of $\mathbf{x}$ are greater than or equal to -1 . The second tests whether all of the elements of $\mathbf{x}$ are less than or equal to 1 . The logical AND operator ensures that the SQRT function is called only if both of the conditions are satisfied. The parentheses are not necessary, but are included for clarity. For this example, both conditions are satisfied, so the SQRT function is called. The argument to the SQRT function uses the elementwise power operator (\#\#) to square each element of $\mathbf{x}$. Consequently, the $i$ th element of y is equal to $\sqrt{1-x_{i}^{2}}$, where $x_{i}$ is the $i$ th element of $\mathbf{x}$.

You can often write a logical condition in multiple ways. For example, the following IF-THEN statements are logically equivalent to the previous IF-THEN statement:

```
/* different ways to compute the same logical condition */
if (x< -1) | (x>1) then
    y = "Unable to compute the function.";
else
    y = sqrt(1-x##2);
if ^(x< -1) & ^(x>1) then
    y = sqrt(1-x##2);
else
    y = "Unable to compute the function.";
```

Some programming languages support a feature known as minimal evaluation or short-circuit evaluation. In these languages, the second argument of a logical AND or OR expression is evaluated only if the first evaluation does not already determine the truth of the expression. For example, in the expression $r \& s$, if $r$ contains a zero, then the entire expression is false, regardless of the value of $\mathbf{s}$. Similarly, in the expression $\boldsymbol{r} \mid \mathbf{s}$, if $\boldsymbol{r}$ contains all nonzero values, then the entire expression is true. The SAS/IML language does not support short-circuit evaluation.

Programming Tip: The SAS/IML language does not support short-circuit evaluation.

Why does this matter? Some programmers write logical expressions that assume short-circuiting will occur. For example, in some languages you can write the following logical expression:

```
/* incorrect: second condition evaluated even if first is false */
if (x>0) & (log(x)>1) then do; /* ERROR if x<=0 */
    /* more statements */
end;
```

This is a valid expression in a language that short-circuits operations because the expression that contains $\log (\mathbf{x})$ is executed only if $\mathbf{x}$ is positive, as ensured by the first condition. However, in the SAS/IML language, both expressions are evaluated, and then the AND operator combines the results. This means that the LOG function is always called and that the program will stop with an error if some element of $\mathbf{x}$ is nonpositive.

Instead, you need to write the previous statements by using nested IF-THEN statements, as shown in the following example:

```
/* correct: second condition not evaluated if first is false */
if x>0 then
    if log(x)>1 then do;
        /* more statements */
    end;
```


### 2.11 Operations on Sets

There are three SAS/IML functions that perform operations on sets. The UNION function returns the union of the values in one or more matrices. The XSECT function returns the intersection of the values in two or more matrices. (The intersection is the set of values common to all matrices.) The SETDIF function accepts two arguments, say $\mathbf{A}$ and $\mathbf{B}$, and returns the values in matrix $\mathbf{A}$ that are not found in matrix $\boldsymbol{B}$.

These set functions are shown in the following statements:

```
/* union, intersection, and difference of sets */
A = 1:4;
B = do(2.5, 4.5, 0.5);
u = union(A, B); /* union */
inter = xsect (A, B) ; /* intersection */
dif = setdif(A, B); /* difference between sets */
print u, inter, dif;
```

As shown in Figure 2.41, each of these functions returns a row vector. The vector u contains the union of the values in $\mathbf{A}$ and B , the vector inter contains the itersection of the values, and dif contains the difference between the sets determined by $\mathbf{A}$ and $\mathbf{~}$.

Figure 2.41 Results of Operations on Sets


In this example, the intersection and the difference between sets were both nonempty, but this is not always the case. The XSECT function returns an empty matrix if there are no elements in the intersection of the matrices. Similarly, the SETDIF function returns an empty matrix if the elements of A are a proper subset of the elements of B. Consequently, you need to check that the matrix returned by XSECT or SETDIF is nonempty before you print it or use it in further computations.

The following statements show an example in which the intersection is empty:

```
/* check whether intersection and set difference are empty */
A = 1:4;
B = do(5, 8, 0.5);
inter = xsect (A, B);
if ncol(inter)>0 then
    print inter;
dif = setdif(A, B);
if ncol(dif)>0 then
    print dif;
```

Figure 2.42 A Nonempty Set Difference

|  | dif |  |  |  |
| :---: | :---: | :---: | :---: | :---: |
|  | 1 | 2 | 3 | 4 |

The matrix inter is created by the XSECT function, but it has zero rows and zero columns. You can therefore use the NCOL function to detect the empty intersection. (You could also check to see if type (inter) has the value 'U'.) The program avoids run-time errors by only executing the PRINT statement for nonempty matrices.

Programming Tip: The intersection of sets can be empty, so always check the matrix returned by XSECT and handle the possibility that the matrix is empty. Do the same for set differences found by using the SETDIF function.

### 2.12 Matrix Operators

The fundamental unit in the SAS/IML language is the matrix. This section describes how to perform arithmetic operations that involve scalars, vectors, and matrices.

### 2.12.1 Elementwise Operators

The elementwise operators in the SAS/IML language consist of one unary operator and several binary operators. The unary operator is the negation operator ( - ), which is equivalent to multiplication by -1 . The elementwise binary operators are the addition operator ( + ), the subtraction operator ( - ), the elementwise multiplication operator (\#), the division operator ( $/$ ), and the power or exponentiation operator (\#\#).

In most cases, the SAS/IML language enables you to write concise, high-level expressions that combine scalars, vectors, and matrices. The SAS/IML software determines if it can make sense
of an arithmetic expression in which the matrices have different dimensions. For example, the following statements show how you can write an arithmetic expression that contains a matrix and also a scalar or a vector:

```
/* elementwise matrix operations with scalar or vector */
x = {7 7, 8 9, 7 9, 5 7, 8 8};
grandmean = 7.5; /* mean of all elements */
y = x - grandmean; /* subtract 7.5 from each element */
mean = {7 8}; /* mean of each column */
xc = x - mean; /* subtract (7 8) from each row */
print y xc;
```

The assignment statement for y contains an expression that subtracts a scalar value from a $5 \times 2$ matrix, $\mathbf{x}$. The SAS/IML language assumes that this is shorthand notation for subtracting the scalar 7.5 from every element of $\mathbf{x}$. On the assignment statement for $\mathbf{x c}$, the program subtracts a $1 \times$ 2 vector from $\mathbf{x}$. Since SAS 9.2, the SAS/IML language has assumed that this is shorthand for subtracting the vector $\{78\}$ from every row of $\mathbf{x}$. The resulting matrices are shown in Figure 2.43.

Figure 2.43 Results of Elementwise Operations

| $y$ |  | $x c$ |  |
| ---: | ---: | ---: | ---: |
| -0.5 | -0.5 | 0 | -1 |
| 0.5 | 1.5 | 1 | 1 |
| -0.5 | 1.5 | 0 | 1 |
| -2.5 | -0.5 | -2 | -1 |
| 0.5 | 0.5 | 1 | 0 |

In general, if m is an $n \times p$ matrix, you can perform elementwise operations with a second matrix $\mathbf{v}$ provided that $\mathbf{v}$ has one of the following dimensions: $1 \times 1, n \times 1,1 \times p$, or $n \times p$. The result of the elementwise operation is shown in Table 2.1, which describes the behavior of elementwise operations for the,+- , \#, /, and \#\# operators.

Table 2.1 Behavior of Elementwise Operators (SAS/IML 9.2)

| Size of v | Result of mop v |
| :---: | :---: |
| $1 \times 1$ | v applied to each element of m |
| $n \times 1$ | v [i] applied to row m[i, ] |
| $1 \times p$ | v [j] applied to column m[, $\mathrm{j}^{\text {] }}$ |
| $n \times p$ | $v[i, j]$ applied to element m[i,j] |

Programming Tip: If you follow the rules presented in Table 2.1, you can perform elementwise operations on rows or columns of a matrix by specifying an appropriate column vector or row vector.

The following statements (which continue from the previous example) give concrete examples for the remaining rules presented in Table 2.1:

```
/* elementwise matrix operations with vector or matrix */
/* r is row vector */
r = {1.225 1}; /* std dev of each col */
std_x = xc / r; /* divide each column (normalize) */
/* c is column vector */
c = x[,1]; /* first column of x */
y = x - c; /* subtract from each column */
/* m is matrix */
m}={6.5 7.5, 7.9 9.1, 7.5 8.5, 5.6 6.4, 7.5 8.5};'
deviations = x - m; /* difference between matrices */
print std_x y deviations;
```

The program shows that you can multiply or divide every row (or column) of a matrix by a scalar or a vector in a natural way. For example, the statement that assigns the matrix std_x divides each column of $\mathbf{x}$ by its standard deviation, thus normalizing the scale of each column. The results of this and the other computations are shown in Figure 2.44.

Figure 2.44 More Results of Elementwise Operations

| std_x | $\mathbf{y}$ | deviations |  |  |  |
| ---: | ---: | ---: | ---: | ---: | ---: | ---: |
| 0 | -1 | 0 | 0 | 0.5 | -0.5 |
| 0.8163265 | 1 | 0 | 1 | 0.1 | -0.1 |
| 0 | 1 | 0 | 2 | 0.5 | 0.5 |
| -1.632653 | -1 | 0 | 2 | -0.6 | 0.6 |
| 0.8163265 | 0 | 0 | 0 | 0.5 | -0.5 |

### 2.12.2 Matrix Computations

The SAS/IML language supports operators for specifying high-level matrix operations such as matrix multiplication. The matrix multiplication operator is *. If $\mathbf{A}$ is an $n \times p$ matrix and $\mathbf{B}$ is a $p \times m$ matrix, the product $\mathbf{A} * \mathbf{B}$ is an $n \times m$ matrix. The $i j$ th entry of the product is the sum $\Sigma_{k=1}^{p} A_{i k} B_{k j}$. If $\mathbf{B}$ is a column vector, then set $j=1$ in the previous formula.

For example, the following statements multiply a matrix and a vector:

```
/* matrix multiplication */
A = {7 7, 8 9, 7 9, 5 7, 8 8}; /* 5 x 2 matrix */
v = {1, -1}; /* 2 x 1 vector */
y = A*v; /* result is 5 x 1 vector */
print y;
```

Figure 2.45 Result of Matrix Multiplication

| $\mathbf{y}$ |  |
| :--- | :--- |
|  | 0 |
|  | -1 |
|  | -2 |
|  | -2 |
| 0 |  |

As a convenience, you can also use the * operator to multiply a matrix by a scalar quantity, such as $3 * \mathbf{A}$. This performs elementwise multiplication and is equivalent to $\mathbf{3 \# A}$.

Another matrix operator is the transpose operator ( ${ }^{`}$ ). This operator is typed by using the GRAVE ACCENT key. (The GRAVE ACCENT key is located in the upper left corner on US and UK QWERTY keyboards.) The operator transposes the matrix that follows it. This notation mimics the notation often seen in statistical textbooks that discuss matrix equations.

For example, given an $n \times p$ data matrix $X$ and a vector of $n$ observed responses, $y$, a goal of ordinary least squares (OLS) regression is to find the linear combination of the columns of $X$ that best approximates $y$. The so-called normal equations provide a computational solution to this problem. The normal equations are written as $\left(X^{\prime} X\right) b=X^{\prime} y$. Solving this equation means solving for the unknown $p \times 1$ parameter vector $b$.

The following statements use the matrix transpose operator to compute the $X^{\prime} X$ matrix and the $X^{\prime} y$ vector for example data:

```
/* Set up the normal equations (x`x)b = x`y */
x = (1:8)`; /* x data: 8 x 1 vector */
y = {5 9 10 15 16 20 22 27}`; /* corresponding Y data */
/* Step 1: Compute X`X and X`y */
x = j(nrow(x), 1, 1) || x; /* add intercept column */
xpx = x` * x; /* cross-products */
xpy = x` * y;
```

The vector $\mathbf{x}$ is initially defined as the column vector that results from transposing the row vector $1: 8$. The vector $y$ is also defined by transposing a row vector. (You can equivalently define $y=\{5$, $9, \ldots, 27\}$ by typing commas between each pair of values, but it is easier to type one grave accent than to type seven commas.) The statements next append a vector of ones to the $\mathbf{x}$ data; this column is used to compute an intercept term in an OLS regression model. The matrix $\mathbf{x p x}$ and the vector xpy are computed by using a natural notional that mimics the mathematics of the problem.

Because the transpose operator can be difficult to see, some programmers prefer to use the T function to indicate transposition. However, the SAS/IML language optimizes certain computations such as the computation of $\mathbf{x p x}$. For that reason, using the transpose operator can sometimes result in better performance than using the T function.

Programming Tip: The SAS/IML language optimizes certain computations that involve the transpose operator (`). Consequently, you should use the transpose operator instead of the T function when you are concerned about the speed of matrix computations.

The SAS/IML language also provides functions that enable you to numerically solve the normal equations for the unknown parameter $b$. Textbooks often write the solution of the normal equations as $b=\left(X^{\prime} X\right)^{-1} X^{\prime} y$, where $A^{-1}$ indicates the inverse of the matrix $A$. The SAS/IML language provides the INV function for computing the inverse of a function, so many SAS/IML programmers use the following statements to numerically solve the normal equations:

```
/* solve linear system */
/* Solution 1: compute inverse with INV (inefficient) */
xpxi = inv(xpx); /* form inverse crossproducts */
b = xpxi * xpy; /* solve for parameter estimates */
```

A better technique is to use the SOLVE function to numerically solve the normal equations:

```
/* Solution 2: compute solution with SOLVE. More efficient */
b = solve(xpx, xpy);
    /* solve for parameter estimates */
```

Explicitly solving for the inverse matrix is inefficient if you are only interested in solving a linear equation. Not only does the INV function need to allocate memory, but the INV function (which solves for a general solution) is often less accurate than the SOLVE function (which solves for a particular solution). There might occasionally be situations in which you need to compute the inverse matrix, but you should avoid it when you have the option.

Programming Tip: If you need to solve the linear equation $A b=z$, use the SOLVE function ( $\mathbf{b}=\mathbf{s o l v e}(\mathbf{A}, \mathbf{z})$ ) unless there is a compelling reason to explicitly compute the inverse matrix $A^{-1}$.

### 2.13 Managing the SAS/IML Workspace

The SAS/IML language enables you to view and manage the matrices that are kept in memory. At any point in your program, you can see the matrices that are defined, free the memory that is associated with matrices, and store matrices to a SAS catalog.

You can use the SHOW statement to display the names, dimensions, and type of matrices that are defined and are in scope. You can specify a list of matrices or you can use the SHOW NAMES statement to display information about all matrices. Both of these techniques are shown in the following statements:

```
/* display the names, dimensions, and type of matrices */
proc iml;
x = 1:3; /* define some matrices */
y = j(1e5, 100); /* large matrix: 10 million elements */
animals = {"Cat" "Dog" "Mouse",
    "Cow" "Pig" "Horse"};
show names; /* show information about all matrices */
show y animals; /* show information about specified matrices */
```

Figure 2.46 Information about Defined Matrices


If you no longer need certain matrices, you can use the FREE statement to free the associated memory. For example, the following statements free the memory that is associated with a large matrix:

```
/* delete one or more matrices by listing their names */
free y;
show names;
```

Figure 2.47 Deleting a Matrix

| SYMBOL | ROWS | COLS TYPE | SIZE |
| :--- | ---: | ---: | ---: |
| ---- | 2 | 3 char | 5 |
| animals | 2 | 3 num | 8 |
| $x$ | 1 | 3 | (includes those without values) |

You can free all matrices by placing a slash (/) on the FREE statement, as shown in the following example. Figure 2.48 shows that three matrix names (symbols) have been used in this session, but none of the matrices have values after the FREE statement is executed.

```
/* delete all matrices */
free /;
show names;
```

Figure 2.48 Deleting All Matrices

```
SYMBOL ROWS COLS TYPE SIZE
```

SYMBOL ROWS COLS TYPE SIZE
Number of symbols = 3 (includes those without values)

```
    Number of symbols = 3 (includes those without values)
```

If you need to shut down your computer or exit the SAS System, but you want to save the state of your program so that you can resume work at a later time, you can save some or all of the matrices that are currently defined. This is especially important if some matrices are the result of a lengthy computation. You can use the STORE statement to save matrices. By default, the STORE statement saves the matrices to a SAS catalog in the Work library. However, the Work library vanishes at the
end of a SAS session, so it is best to use the RESET STORAGE statement to store the matrices to a permanent storage location, as shown in the following example. Figure 2.49 shows that the animals and $\mathbf{x}$ matrices are saved in the MyStorage catalog.

```
/* store all matrices to a permanent location */
x = 1:3;
animals = {"Cat" "Dog" "Mouse", "Cow" "Pig" "Horse"};
libname MyLib "C:\My Data"; /* set directory for storage */
reset storage=MyLib.MyStorage; /* set the storage catalog */
store _all_; /* store all matrices */
show storage; /* display catalog contents */
```

Figure 2.49 Stored Matrices

```
Contents of storage library = MYLIB.MYSTORAGE
Matrices:
ANIMALS X
Modules:
```

When you are ready to resume your work, you can start a new SAS/IML session and use the LOAD statement to restore the matrices, as shown in the following statements:

```
/* load matrices from a storage catalog */
proc iml;
reset storage=MyLib.MyStorage;
load _all_;
show names;
```

Figure 2.50 Loading Matrices from Storage


When you no longer need the matrices that are stored in the SAS catalog, you can remove them by using the REMOVE statement, as shown in the following statements:

```
/* remove all contents from a storage catalog */
remove _all_;
show storage; /* show the contents of the catalog */
```

Figure 2.51 Removing Matrices from Storage

```
Contents of storage library = MYLIB.MYSTORAGE
Matrices:
Modules:
```

As an alternative to using the _ALL_ keyword, you can specify a list of names on the STORE and LOAD statements to indicate which matrices you want to store and load. Similarly, you can remove a specified list of matrices from storage.

From Statistical Programming with SAS/IML® Software by Rick Wicklin. Copyright © 2010, SAS Institute Inc., Cary, North Carolina, USA. ALL RIGHTS RESERVED.


## Index

## Symbols

() (parentheses), 117

* (matrix multiplication), 49
+ (plus sign)
as addition operator, 47
as string concatenation operator, 24
as subscript reduction operator, 66, 81
- (minus sign)
as subtraction operator, 47
- (negation operator), 47
/ (slash)
as division operator, 47
in FREE statement, 52
// (vertical concatenation operator), 42
: (index creation operator), 25
: (subscript reduction operator), 67, 81
; (semicolon) in statements, 5
$<$ (less than operator), 36
<= (less than or equals operator), 36
= (equal sign), 36
$>$ (greater than operator), 36
$>=$ (greater than or equals operator), 36
@ symbol, 100
\# (multiplication operator), 47
\#\# (exponentiation operator), 47
\$ (dollar sign), 254
\& (ampersand)
as AND operator, 43, 62
in vector name, 93
variables and, 103, 271
- (transpose operator), 50, 73
^(NOT operator), 43
^ $=$ (not equals operator), 36
|| (horizontal concatenation operator), 41
I (OR operator), 43


## A

abstract classes, 134
action menu
attaching to graphs, 386
methods supporting, 139, 429
addition operator (+), 47
AICC (Akaike's information criterion), 217, 267, 390
aicc function (R), 271
algorithm performance
comparing, 375
timing algorithms, 383
ALIAS statement, 118
aliases
creating for modules, 117
defined, 114
ALL function, 422
ALL option, READ statement, 56
ampersand (\&)
as AND operator, 43, 62
in vector name, 93
variables and, 103, 271
AND operator (\&), 43, 62
ANY function, 422
APPEND statement
description, 425
timing algorithms, 383
transferring data, 92
writing data sets, 58
arguments
evaluating, 75
functions requiring, 103
optional, 207
passing by reference, 74
to modules, 331
Auto Hide property, 294
Auxiliary Input window, 124

## B

bar charts
creating, 135
creating from data objects, 146
creating from vectors, 145
defined, 145
modifying graphs, 146
BarChart class
creating graphs, 135, 145
frequently used methods, 147, 432
modifying graphs, 146
base classes, 134, 141
birthday matching problem
case study, 338
overview, 332
simulating, 335
BlendColors module, 120, 236
bootstrap distributions
computing estimates, 353
creating, 351
defined, 349
for the mean, 350
plotting, 368
bootstrap methods
case study, 363
comparing two groups, 356
defined, 349
overview, 349
box plots
categorical data and, 193
creating, 162
creating from data objects, 163
creating from vectors, 163
creating grouped, 164
defined, 161
BoxPlot class
creating graphs, 161
frequently used methods, 165, 432
break points, 232
BY group processing, 71
BY statement, 362, 365

## C

C functions, 346
C = option, HISTOGRAM statement
(UNIVARIATE), 95
case sensitivity, 5, 130, 254
categorical variables
changing display order of, 236
displaying on graphs, 193
indicating values via marker shapes, 226
CDF function, 421
CEIL function, 312, 397
central limit theorem, 355
CHAR function, 71, 424
character matrix
length of, 22
overview, 18, 21
CHOOSE function, 65, 319, 422
CLASS statement, 289
classes, see IMLPlus classes
classification variables
creating line plots with, 158
regression diagnostics and, 289
client data sets, 174
CLOSE statement, 425
CMISS function, 77
coin-tossing simulation, 312
COLNAME $=$ option
FROM clause, CREATE statement, 58
PRINT statement, 19, 36, 57, 71, 426
colon operator (:), 25
color interpolation, 234
color ramp, 119, 232
ColorCodeObs module, 120, 236
ColorCodeObsByGroups module, 120, 236
column vectors
extracting data from matrices, 32
overview, 18, 28
columns, standardizing in matrix, 83
COLVEC module, 331
comparison operators, 36
CONCAT function, 24, 97, 100
concatenation operators
overview, 41, 97
padding strings and, 24
concrete classes, 134
conditioning plot, 274
confidence intervals, plotting, 366
constant matrices, 24
continuous variables
coloring by values, 232
marker colors to indicate values, 229
ContourPlot class, 168
control statements, see also DO statement, see also IF-THEN/ELSE statement
overview, 38, 425
syntax for, 5
Cook's $D$ statistic, 284, 292
coordinate systems
drawing on graphs, 191
practical differences, 202
CopyServerDataToDataObject module, 119, 184, 299
CORR procedure, 406, 408
correlation analysis, 408
COUNTMISS function, 422
COUNTN function, 67, 422
COV option, CORR procedure, 406, 408
craps (game) simulation, 322
CREATE statement
description, 425
timing algorithms, 383
transferring data, 92
writing data sets, 58,113
CUSUM function, 422
cutoff values, 232

## D

data attributes
changing category display order, 236
changing marker properties, 226
getting/setting, 244
R functions, 256
selecting observations, 241
data frames
creating, 256
defined, 254
passing data to R, 254
R objects and, 260
data objects
adding variables to, 180
creating, 132, 174
creating bar charts from, 146
creating box plots from, 163
creating data sets from, 177
creating from data sets, 174
creating from matrices, 177
creating histograms from, 150
creating line plots from, 156,160
creating linked graphs from, 175
creating matrices from, 179
creating scatter plots from, 154
defined, 144
data sets
adding variables from, 184
Birthdays2002, 417
creating data objects from, 174
creating from data objects, 177
creating from matrices, 58
creating from tables, 439
creating matrices from, 56
linking related data, 390
Movies, 416
reading/writing, 56, 113, 425
server/client, 174
transferring data from R, 257
transferring to R functions, 254
Vehicles, 415
DataObject class
adding variables to data objects, 180
categorical variable display order, 236
changing marker properties, 226
class hierarchy, 135
creating data objects, 174,177
creating data sets, 177
creating graphs, 144
creating linked graphs, 175
creating matrices, 179
creating R data frames, 256
frequently used methods, 243, 259, 428
getting/setting attributes, 244
overview, 112, 131, 173, 185, 428
schematic description of roles, 132
selecting observations, 241
transferring data from R, 257
DATASETS procedure, 102
DataTable class, 138
DataView class
class hierarchy, 134
creating graphs, 136
displaying action menu items, 387
frequently used methods, 429
linking related data, 391
overview, 139, 429
DATE7. format, 156, 170
DATETIMEw. format, 256
DATEw. format, 256
debugging programs
Auxiliary Input window and, 124
jumping to error location, 121
jumping to errors in modules, 123
PAUSE statement and, 125
declare keyword, 130, 144
default module storage directory (DMSD), 114
derived classes, defined, 134
descriptive statistical functions, 421
DET function, 423
DIAG function, 33, 423
diagonal of a matrix, 33
dialog boxes
creating with Java, 402
creating with R, 404
displaying lists in, 399
displaying simple, 396
in SAS/IML Studio, 126, 396
modal, 405
division operator (/), 47
DMSD (default module storage directory), 114
DO function, 25, 377, 423
DO statement, 5, 38, 39, 425
DO/UNTIL statement, 40, 425
DO/WHILE statement, 40, 425
DoDialogGetDouble module, 120
DoDialogGetListItem module, 120, 399
DoDialogGetListItems module, 120, 399, 401
DoDialogGetString module, 120
DoDialogModifyDouble module, 120, 397
DoErrorMessageBoxOK module, 120
dollar sign (\$), 254
DoMessageBoxOK module, 120, 399
DoMessageBoxYesNo module, 120, 397
DrawContinuousLegend module, 120
drawing regions
drawing on graphs, 191
foreground/background, 197
predication band case study, 198
drawing subsystem, 187
DrawInset module, 120, 206
DrawLegend module
adjusting graph margins, 208
comparing regression models, 300
description, 120
overview, 204
DrawPolygonsByGroups module, 120
dynamically linked graphs, 136
E
ECDF (empirical cumulative distribution function), 287
EIGEN subroutine, 423
elementwise operators, 47
empirical cumulative distribution function (ECDF), 287
empty matrix, 64, 207
ENDSUBMIT statement, see SUBMIT statement equal sign (=), 36
error handling when calling procedures, 101
error messages, interpretting, 31
EXECUTE subroutine, 104
ExecuteOSProgram module, 308
exponentiation operator (\#\#), 47
ExportDataSetToR module, 254, 256
ExportMatrixToR module, 255
ExportToR module, 256

## F

FINISH statement, 72, 424
fitted function (R), 262
FLOOR function, 312
FORMAT procedure, 247
FORMAT = option, PRINT statement, 19, 426
formatting functions, 424
FREE statement, 42, 52
Freedman-Diaconis bandwidth, 151
FREQ procedure, 71
FROM clause, APPEND statement, 58
FROM clause, CREATE statement, 58
function modules, 72, 270
functions
as methods, 130, 144
built-in, 435
calling in R packages, 264
creating matrices, 24
descriptive statistical, 421
finding minimum of, 84
formatting, 424
linear algebra, 423
mathematical, 420
matrix query, 422
matrix reshaping, 422
probability, 421
requiring list arguments, 103
set, 46, 423
unimodal, 84

## G

GAM procedure, 382
GCV (generalized cross validation), 217
generic functions, 260
GENMOD procedure, 358
GetPersonalFilesDirectory module, 119
GINV function, 423
GLM procedure
comparing models, 297
MODEL statement, 98, 290
naming output variables, 96
OUTPUT statement, 96, 189, 232
regression curves on scatter plot, 188
running regression analysis, 280
variables for predicted/residual values, 182
GLOBAL clause, START statement, 74, 117
global symbols, 74
golden section search, 84
graphs, see IMLPlus graphs, see statistical graphs
greater than operator ( $>$ ), 36
greater than or equals operator $(>=), 36$
grouped box plots, 164
GTK2 toolkit, 404
gWidgets package (R), 404

## H

HARMEAN function, 103
high-leverage observations, 286
hist function (R), 273
Histogram class
creating graphs, 149
frequently used methods, 150,432
HISTOGRAM statement, UNIVARIATE procedure, 95
histograms
adding rug plots, 212
creating from data objects, 150
creating from vectors, 149
defined, 149
KDE case study, 215
horizontal concatenation operator (||), 41

## I

I function, 34, 423
identity matrix, 34
IF-THEN/ELSE statement
description, 38, 425
handling errors, 102
logical operators and, 45
syntax for, 5
IML procedure
calling R functions, 252
drawing comparison, 224
global scope and, 74
overview, 6
RUN statement and, 6
running, 7
timing algorithms, 383
IMLMLIB module library, 73, 78, 119
IMLPlus
calling R functions, 112, 252, 434
calling SAS procedures, 90, 111, 433
color representation in, 229
debugging programs, 121
drawing comparison, 224
managing data, 112, 173
object-oriented programming, 130
overview, 109, 427
processing requirements, 8
SAS/IML comparison, 126
IMLPlus classes
base classes, 134
creating graphs, 135
DataObject class, 131, 428
DataView class, 429
defined, 144
derived classes, 134
in modules, 141
overview, 129
Plot class, 429
viewing documentation, 133
IMLPlus graphs
adding lines to, 210
adjusting margins, 208
attaching menus to, 386
changing axis tick positions, 220
characteristics, 112
coordinate systems, 191
creating, 135
creating dynamically linked, 136
drawing figures/diagrams, 222
drawing in background, 197
drawing in foreground, 197
drawing legends, 204
drawing regions, 191
drawing rug plots, 212
drawing subsystem, 187
KDE case study, 214
methods for creating/modifying, 431
methods for drawing, 430
plotting loess curve, 216
R functions and, 273
scatter plot case study, 198
IMLPlus modules
base classes in, 141
debugging, 123
frequently used, 119
local variables in, 117
overview, 114
passing objects to, 139
storing/loading, 114
import keyword, 402
ImportDataSetFromR module, 258
ImportMatrixFromR module, 258
index creation operator (:), 26
index of occurrence, 338
INFLUENCE option, MODEL statement (REG), 284
influential observations, 284
INSERT function, 43, 423
insets, drawing on graphs, 203
interactive techniques
attaching menus to graphs, 386
dialog boxes and, 396
linking related data, 390
pausing programs, 385
INTERP $=$ option, MODEL statement (LOESS), 268
INTO clause, READ statement, 57
IntToRGB module, 120, 230, 236
INV function, 50, 372, 423
IRR function, 104

## J

J function, 24, 70, 423
Java language, 130, 402
javax.swing classes, 402
JOptionPane class (Java), 402

## K

kernel density estimate case study, 94,214
KERNEL option, HISTOGRAM statement
(UNIVARIATE), 95
KernSmooth package (R), 264

## L

LABEL= option, PRINT statement, 19, 36, 426
LCLM = option, OUTPUT statement (GLM), 96
least squares regression model, 296
LEFT function, 424
legends, drawing on graphs, 203
LENGTH function, 22, 422
less than operator (<), 36
less than or equals operator $(<=), 36$
LIBNAME statement, 59
library function (R), 264
line plots
creating for several variables, 156
creating for single variable, 155
creating from data objects, 156,160
creating from vectors, 155
creating with classification variables, 158
defined, 155
line segments, drawing, 191
linear algebra functions, 423
LinePlot class, 155, 433
frequently used methods, 161
lines function (R), 273
linked graphs, 136, 175
lists, displaying in dialog boxes, 399
lm function (R), 259
LOAD statement
description, 424
loading matrices, 53
loading modules, 114
storing modules, 76
LOC function
analyzing observations by categories, 68
assigning values to observations, 65
description, 422
efficient programs and, 79
handling missing values, 67
locating observations, 60
local variables, 73,117
loess curve case study, 216
loess function (R), 267
LOESS procedure
calling, 382
linking related data, 392
MODEL statement, 268, 392
plotting loess curve, 216
SCORE statement, 217
SELECT= option, 267
logical operators, 43
LOGISTIC procedure, 304
logistic regression diagnostics, 303
1s function (R), 255

## M

macro variables, 99
MAD (median absolute deviation), 4
MAD function, 4
margins, graph, 208
marker properties, changing, 226
mathematical functions, 420
matrix multiplication operator (*), 49
matrix operators, 47
matrix query functions, 422
matrix reshaping functions, 422
matrix transpose operator (`), 50, 73
matrix/matrices
changing shape of, 29
combining, 47
computing, 49
creating, 18
creating data objects from, 177
creating data sets from, 58
creating from data objects, 179
creating from data sets, 56
creating macro variables from, 99
defined, 4
diagonals of, 33
dimensions of, 20
empty, 64, 207
extracting data from, 30
length of, 22
options for printing, 425
passing data to R functions, 254
printing, 19
standardizing columns in, 83
transferring data, 91
transposing, 28
types of, 21
using functions to create, 24
MATTRIB statement, 57
MAX function, 422
MEAN function
description, 84, 422
subscript reduction operators and, 67, 82
mean function (R), 261
Mean module, 435
MEANS procedure, 361, 363
median absolute deviation (MAD), 4
Median module, 4
menus, attaching to graphs, 386
Mersenne-Twister random number generator, 27
methods
action menu and, 139, 429
calling syntax, 130
defined, 130, 144
for creating/modifying graphs, 431
for drawing graphs, 430
related to action menu, 429
MI procedure, 66
MIANALYZE procedure, 66
MIN function, 422
minimal evaluation, 45
minus sign (-)
as subtraction operator, 47
missing values
algorithms that delete, 375
comparison operators and, 37
handling, 65
passing, 207
R functions and, 261
standardizing data with, 84
modal dialog boxes
defined, 405
for correlation analysis, 408
with checkboxes, 406
MODEL statement
GLM procedure, 98, 290
LOESS procedure, 268, 392
REG procedure, 284
modules, see also IMLPlus modules, see also specific modules
aliases for, 114,117
arguments to, 331
defined, 424
dialog boxes and, 126
drawing rug plots on graphs, 212
encapsulating R statements, 270
evaluating arguments, 75
finding minimum of functions, 84
function, 72
global symbols, 74
IMLMLIB library, 73, 78
local variables, 73
overview, 435
passing arguments by reference, 74
storing, 76
subroutine, 72
MONNAMEw. format, 236
MosaicPlot class, 168
multiplication operator (\#), 47

## N

names function (R), 254, 263
NCOL function
analyzing observations by categories, 68
AND operator and, 62
description, 422
determining matrix dimensions, 20
empty matrix and, 64
operations on sets, 47
negation operator (-), 47
NLENG function, 22,422
NOMISS option, CORR procedure, 409
NORMAL function, 27, 421
NOSIMPLE option, CORR procedure, 408, 410
not equals operator ( $\wedge=$ ), 36
NOT operator ( ${ }^{\wedge}$ ), 43
NROW function, 20, 64, 422
NUM function, 424
numeric matrix, 21

0
object-oriented programming, 130
objects
defined, 144
passing to modules, 139
observations
analyzing by categories, 68
assigning values to, 65
attributes of, 246
identifying high-leverage, 286
identifying influential, 284
locating, 60
selecting, 241
ODS EXCLUDE statement, 438
ODS GRAPHICS statement, 440
ODS HTML statement, 307
ODS OUTPUT statement, 92, 439
ODS SELECT statement, 438
ODS statements
creating statistical graphics, 440
finding table names, 437
overview, 437
selecting/excluding tables, 438
ODS statistical graphics, 307, 440
ODS tables, see tables
ODS TRACE statement, 437
OK = option, SUBMIT statement, 101, 111
onCancel function (R), 408
onOK function (R), 408
operators
comparison, 36
concatenation, 24, 41, 97
elementwise, 47
frequently used, 420
logical, 43
matrix, 47
subscript reduction, 66, 81
Options dialog box, 114
options function (R), 263
OR operator (I), 43
OUTHITS option, SURVEYSELECT procedure, 360, 362
outliers
defined, 231
identifying, 286
influential observations and, 284
LOC function and, 63
marking via color, 231
OUTPUT statement
GLM procedure, 96, 189, 232
UNIVARIATE procedure, 94, 101
OutputDocument class, 277, 295

## P

packages (R)
calling functions in, 264
defined, 252
installing, 264
parameters
optimizing smoothing, 267
passing to procedures, 93,111
parentheses (), 117
passing arguments by reference, 74
PAUSE statement, 125, 385
PBSPLINE statement, SGPLOT procedure, 383
PCTLPTS = option, OUTPUT statement
(UNIVARIATE), 94, 101
PDF function, 157, 421
Pearson correlation coefficient, 72
PERCENT6.1 format, 245
performance considerations
avoiding program loops, 79
comparing algorithm performance, 375
subscript reduction operators, 80
Plot class
adjusting graph margins, 208
creating graphs, 135
drawing legends, 204
drawing line segments, 191
drawing subsystem, 187
frequently used methods, 430
graph coordinate systems, 191
overview, 138, 429
plot function (R), 273
Plot2D class, 135, 431
plus sign (+)
as addition operator, 47
as string concatenation operator, 24
as subscript reduction operator, 66, 81
points function (R), 273
PolygonPlot class, 168
power operator (\#\#), 47
prediction bands, scatter plots, 198
PRINCOMP procedure, 363
PRINT statement
COLNAME $=$ option, 19, 36, 57, 71, 426
FORMAT = option, 19, 230, 426
LABEL= option, 19, 36, 426
ROWNAME $=$ option, 19, 71, 426
syntax, 19
printing
expressions, 35
matrices, 19, 425
submatrices, 35
PRINTNOW statement, 297
probability functions, 421
procedures
bootstrap computations, 358
calling from IMLPlus, 90, 111, 433
checking return code, 111
creating macro variables from matrices, 99
functions requiring list arguments, 103
handling errors, 101
kernel density estimate case study, 94
passing parameters, 93, 111
transferring data, 91
PROD function, 82, 422
profiling programs, 372
programming techniques
analyzing observations by, 68
applying variable transformation, 59
assigning values to observations, 65
avoiding loops, 79
finding minimum of functions, 84
handling missing values, 65
locating observations that satisfy criteria, 60
reading/writing data, 56
standardizing columns in matrix, 83
properties, variable, 244
pseudorandom matrices, 27
\%PUT statement, 99
PUTC function, 23, 424
PUTN function, 71, 221, 424

## Q

Q-Q plots, 287
Qntl module, 436
QNTL subroutine, 354, 422
QR subroutine, 423
QUANTILE function, 421
QUARTILE module, 133, 152
QUIT statement, 102

## R

R language
calling functions, 112, 252, 434
calling R packages, 264
creating dialog boxes, 404
creating graphics, 273
data attributes, 256
encapsulating in modules, 270
handling missing values, 261
importing R objects, 259
optimizing smoothing parameter, 267
overview, 252
passing data, 254
R, supported versions, 264
RANDGEN subroutine
arguments to, 403
coin-tossing simulation, 313
description, 421
dialog boxes and, 397
Mersenne-Twister random number generator, 28
outcomes with specified probabilities, 322, 327
rolling-dice simulation, 321
random number generation, 27, 319
random sampling
bootstrap methods and, 350
defined, 312
with unequal probability, 327
RANDSEED subroutine, 28, 58, 312, 321, 421
RANK function, 239, 422
RANKTIE function, 422
READ statement, 56, 425
reading data sets, 56, 113, 425
reference lines, adding, 210
REG procedure, 102, 284
regression curves, overlaying, 188
regression diagnostics
case study, 296
classification variables and, 289
comparing models, 292
displaying lines, 210
examining residuals distribution, 287
fitting regression models, 280
identifying high-leverage observations, 286
identifying influential observations, 284
identifying outliers, 286
logistic, 303
overview, 279
REMOVE function, 43, 375, 423
REMOVE statement, 53
REPEAT function, 25, 423
resampling, 359
RESET STORAGE statement, 53, 77
return code, 111, 397
RETURN statement, 72
RGB coordinate system, 199, 229
RGBToInt module, 120, 230, 236
RIGHT function, 424
robust regression model, 296
ROBUSTREG procedure, 63, 297
rolling-dice simulation, 321
ROOT function, 423
RotatingPlot class, 168
row index, 32
row vectors
extracting data from matrices, 32
overview, 18, 28
standardizing columns in matrix, 83

ROWNAME= option, PRINT statement, 19, 71, 426
ROWVEC module, 330
RSTUDENT = option, OUTPUT statement (GLM), 232
rug plots, 212
RUN statement, 6

## S

sampling
overview, 311
with replacement, 329
with unequal probability, 327
SAS Foundation, 9
SAS Metadata Server Connection Wizard, 10
SAS procedures, see procedures
SAS/IML language, 3, see also IMLPlus, 18, 419
SAS/IML modules, see modules
SAS/IML software, see IML procedure, see SAS/IML Studio
SAS/IML Studio, see also IMLPlus
calling C functions, 346
client-server architecture, 9
dialog boxes in, 126, 396
exploratory data analysis, 12
installing, 10
invoking, 10
online help, 432
overview, 8
pasting graphs, 277
running programs in, 11
workspace example, 8
scalars, 18,47
scatter plots
adding rug plots, 212
categorical data and, 193
comparing analyses with, 295
creating, 427
creating from data objects, 154
creating from vectors, 154
defined, 153
marker shapes and, 226
modifying, 427
overlaying regression curves, 188
prediction band case study, 198
ScatterPlot class, 153, 427, 433
scree plot, 364, 366
seed value, 27
SELECT = option, LOESS procedure, 267
SEM (standard error of the mean), 353
semicolon (;), 5
server data sets, 174
set functions, 46, 423

SETDIF function, 32, 46, 424
SGPLOT procedure, 383
SHAPE function, 29, 423
short-circuit evaluation, 45
SHOW NAMES statement, 51, 73
SHOW statement, 51
signature, defined, 130
simulation
birthday matching problem, 335, 341
coin tossing, 312
defined, 312
efficiency considerations, 319
game of craps, 322
rolling dice, 321
slash (/)
as division operator, 47
in FREE statement, 52
SMOOTH = option, MODEL statement (LOESS), 392
smoothing parameter, optimizing, 267
SOLUTION option, MODEL statement (GLM), 290
SOLVE function, 373, 423
SORT procedure, 71
SPLINE subroutine, 382
SQRT function, 45
SSQ function, 422
standard error of the mean (SEM), 353
START statement
description, 72, 424
GLOBAL clause, 74, 117
parentheses on, 117
statements, see also ODS statements
@ symbol and, 100
control, 425
module, 424
reading/writing data sets, 425
semicolon in, 5
statistical graphs, see also bar charts, see also box plots, see also histograms, see also IMLPlus graphs, see also line plots, see also scatter plots
changing graph axis format, 169
data sources for, 144
displaying underlying data, 168
drawbacks of creating from vectors, 172
ODS, 307, 440
overlaying reference lines, 210
summary of graph types, 167
STORE statement
description, 424
saving matrices, 53
storing modules, 76, 115
storing modules, 76, 114
str function (R), 260
string concatenation operator (+), 24
strings, removing blanks in, 98
STRIP function
description, 424
removing blanks, $24,98,100$
submatrices, printing, 35
SUBMIT statement
calling procedures, $90,111,434$
calling R functions, 252, 266
creating macro variables from matrices, 99
description, 425
OK= option, 101, 111
passing procedure parameters, 93
R option, 253
SUBPAD function, 98
subroutine modules, 72
subscript reduction operators, 66,81
subtraction operator (-), 47
SUM function, 66, 422
summary function (R), 254
SURVEYSELECT procedure, 358
SVD subroutine, 423
SYMGET function, 102
SYMGETN function, 102
SYMPUT subroutine, 99
SYSERR macro variable, 102
SYSERRORTEXT macro variable, 102
SYSWARNINGTEXT macro variable, 102

## T

T function, 28, 423
tables
creating data sets from, 439
finding names of, 437
listing destination, 307
regression diagnostics and, 291
selecting/excluding, 438
Tcl (Tool Command Language), 404
tcltk package (R), 404, 406
tclVar function (R), 407
tick marks, changing for date axis, 220
TIME function, 372
TIMEw. format, 256
timing computations
overview, 371
replicating timings, 379
timing algorithms, 383
tips for, 384
tkbutton function (R), 405
tkconfigure function (R), 407
tkfocus function (R), 406
tkgrid function (R), 405, 407
tktitle function (R), 405
tktoplevel function (R), 405
tkwait function (R), 406
Tool Command Language (Tcl), 404
TPSPLINE subroutine, 382
TRACE function, 423
transforming variables, 59, 181, 289
transpose operator (`), 50, 73
TRIM function, 24, 424
TYPE function, 21, 47, 64, 422

## U

UCLM = option, OUTPUT statement (GLM), 96
UNIFORM function, 27, 421
unimodal functions, 84
UNION function, 46, 424
UNIQUE function, 68, 79, 424
UNIVARIATE procedure
calling, 90
comparing quantiles of data, 287
handling errors, 101
KDE case study, 95
transferring data, 93
USE statement, 56, 113, 425

## V

VAR function, 84, 354, 422
Var module, 354, 436
variables, see also specific types of variables
adding for predicted/residual values, 182
adding from data sets, 184
adding to data objects, 180
categorical, 193
creating from matrices, 58
modules and, 73
output, 96
properties of, 244
reading from data sets, 56
transforming, 59, 181, 289
VECDIAG function, 33, 423
vectors
column, 18
combining, 47
creating, 18
creating bar charts from, 145
creating box plots from, 163
creating histograms from, 149
creating line plots from, 155
creating scatter plots from, 154
module arguments as, 331
row, 18
vertical concatenation operator (//), 42

## W

WHERE clause, 113
Windows clipboard, 277
Work library, 8,52
workspace bar, 9
workspaces
comparing analyses in, 293
defined, 8
example, 8
managing, 51
server connections, 10
Work library and, 8,52
writing data sets, $56,113,425$

## X

XSECT function, 46, 424


# Gain Greater Insight into Your SAS ${ }^{\circledR}$ Software with SAS Books. 

Discover all that you need on your journey to knowledge and empowerment.


